#-------------------------------------------------
#-------------------------------------------------

TEMPLATE = lib
CONFIG  += plugin
TARGET   = $$qtLibraryTarget(utausynth)


INCLUDEPATH += ../../editor ../ ../../tools
LIBS += -Wl,--no-undefined

PKGCONFIG += glib-2.0

LIBS += -lsekai -lworld -ljack -lsndfile

CONFIG      += link_pkgconfig


HEADERS += \
    utau_synth.h \
    ../../tools/utauloid/oto.h \
    ../../editor/Utils.h \
    utaudb.h \
    otoini.h

SOURCES += \
    utau_synth.cpp \
    ../../editor/Utils.cpp \
    utaudb.cpp \
    otoini.cpp

QMAKE_CXXFLAGS += -Wall -std=c++11

#--------------------------------------------
CONFIG(debug, debug|release) {
    COMPILEDIR = $${OUT_PWD}/../../debug
    LIBS += -Wl,--no-undefined
} else {
    COMPILEDIR = $${OUT_PWD}/../../release
}

DESTDIR         = $${COMPILEDIR}/plugins
OBJECTS_DIR     = $${COMPILEDIR}/utausynth/.obj
MOC_DIR         = $${COMPILEDIR}/utausynth/.moc
RCC_DIR         = $${COMPILEDIR}/utausynth/.rcc
UI_DIR          = $${COMPILEDIR}/utausynth/.ui
#--------------------------------------------

unix {
    target.path = $${PREFIX}/lib/qtau/plugins
    INSTALLS += target
}
